源氏香図の総数とその一般化について
―― パソコンの利用 ――

田村(たむら) 隆(たかし)

1．はじめに

昨年（1991年）の1学期，本校3年生の「確率・統計」の授業において「場合の数」を扱ったとき，生徒に *p*. 22 のように「源氏香図」と同内容の課題を出してみました．

この別解として，起り得るすべての場合，3125通りをパソコンを用いて調べてみたところ，その一般化が容易であることに考えつきましたので，紹介いたします．

参考文献：
日本国語大辞典　　小学館
順列組合より確率まで　藤森良夫著
東京考へ方研究社

2．源氏香図とは

源氏香とは，後水尾天皇の頃（江戸時代）に考案された組み香のひとつで，5種の香をそれぞれ5包計25包作り，香元が任意に5包ずつとってたき，他の人たちがかぎわけて，その異同を5本の線で示すゲームである．そのルールは，1番目，2番目，3番目，4番目，5番目がみな異なる香と判断すれば解答紙に右側から順に縦線を5本　と書き，また1番目，3番目が同香で2番目，4番目，5番目が同香と判断すれば，右側から対応する縦線を横線で　のように結んで解答するというものである．このような図形は全部で52通りできるので，「源氏物語」54巻中の桐壺と夢浮橋の2巻を除いた各巻名を図形に付けて源氏香図と呼んでいるわけである．(源氏香図については，*p*. 22 参照)

3．源氏香図の数学的説明とパソコンの利用

源氏香図は，数学的に次のように述べることができるであろう．

5種類の数字1，2，3，4，5から同じ数字を繰り返し用いることを許して5桁の数を作る．このとき，

数３５１４２　には　図形

数１１２３２　には　図形

数４５５４４　には　図形

……………………

……………………

というように5桁の数の集合から5本の縦線と幾つかの横線を用いた図形の集合の上への写像を考えることができる．5本の縦線は右から順に，一位，十位，百位，千位，万位の数字に対応し，同じ数字があると対応する縦線の上端が横線で結ばれるわけである．

この場合の図形の総数が52通りであることは，順列組合せの計算によって解答できる例題として知られている（例　藤森良夫著　順列組合より確率まで）．

ここではパソコンを用いて，可能なすべての5桁の数 5^5 個から52個の図形が生ずることを示そう．そのために変数I，J，K，L，Mはそれぞれ1から5までの整数を表し，配列変数 $D(1)$，$D(2)$，$D(3)$，$D(4)$，$D(5)$ は，$1 \leqq D(t) \leqq t$ なる整数を表すものとして次のような対応を考える．($D_t = D(t)$)

（対応方法）

(MLKJI)		(図形)		($D_5\ D_4\ D_3\ D_2\ D_1$)
3 5 1 4 2	⟶		⟶	5 4 3 2 1
1 1 2 3 2	⟶		⟶	4 4 1 2 1
4 5 5 4 4	⟶		⟶	1 3 3 1 1
……………		……		…………………
……………		……		…………………

例えば，数１１２３２では一位と百位が同じ数字だから $D(3)=D(1)=1$，また千位と万位が同じ数字だから $D(5)=D(4)=4$，十位は同じ数字が無い

ので $D(2)=2$ とする．すなわち，$D(t)$ は5桁の数の第 t 桁の数字と同じものが，桁数の小さい方から見て初めて第 s 桁にあるとき，$D(t)=s$，第 t 桁未満にそれと同じ数字が無ければ，$D(t)=t$ となる変数である．このように $D(t)$ を定義すれば，源氏香図と数 $D_5D_4D_3D_2D_1$ を1対1に対応させせることができる．このことをパソコンで実行した結果を下に示そう．

```
10    REM 源氏香図の総数しらべ
20    LPRINT "ﾊﾟｿｺﾝ による源氏香図の総数しらべ"
30    DIM D(5),P(3125),PC(3125)
40    D(1)=1:D(2)=2:D(3)=3:D(4)=4:D(5)=5
50    Q=0:S=0:TS=0
60    FOR I=1 TO 5
70     FOR J=1 TO 5
80      FOR K=1 TO 5
90       FOR L=1 TO 5
100       FOR M=1 TO 5
110        IF J=I THEN D(2)=D(1)
120        IF K=I THEN D(3)=D(1)
130        IF L=I THEN D(4)=D(1)
140        IF M=I THEN D(5)=D(1)
150        IF K=J THEN D(3)=D(2)
160        IF L=J THEN D(4)=D(2)
170        IF M=J THEN D(5)=D(2)
180        IF L=K THEN D(4)=D(3)
190        IF M=K THEN D(5)=D(3)
200        IF M=L THEN D(5)=D(4)
210    Q=Q+1
220    P(Q)=D(1)+D(2)*10+D(3)*100+D(4)*1000+D(5)*10000
230    D(1)=1:D(2)=2:D(3)=3:D(4)=4:D(5)=5
240   NEXT M:NEXT L:NEXT K:NEXT J:NEXT I
250   FOR T=1 TO 3125
260   IF P(T)>0 THEN PC(T)=1 ELSE 300
270    FOR R=T+1 TO 3125
280    IF P(T)=P(R) THEN PC(T)=PC(T)+1:P(R)=0
290    NEXT R
300   NEXT T
310   LPRINT
320   FOR T=1 TO 3125
330   IF P(T)=0 THEN 360 ELSE TS=TS+1
340   LPRINT USING "(###) ###### 型 ..... ###";TS,P(T),PC(T)
350   S=S+PC(T)
360   NEXT T
370   LPRINT  ".............................."
380   LPRINT USING "  計           ###########";S
390   LPRINT USING "  源氏香図総数 =####";TS
400   END
```

```
ﾊﾟｿｺﾝ による源氏香図の総数しらべ

(  1)  11111 型 .....    5
(  2)  51111 型 .....   20
(  3)  14111 型 .....   20
(  4)  44111 型 .....   20
(  5)  54111 型 .....   60
(  6)  11311 型 .....   20
(  7)  31311 型 .....   20
(  8)  51311 型 .....   60
(  9)  13311 型 .....   20
( 10)  33311 型 .....   20
( 11)  53311 型 .....   60
( 12)  14311 型 .....   60
( 13)  34311 型 .....   60
( 14)  44311 型 .....   60
( 15)  54311 型 .....  120
( 16)  11121 型 .....   20
( 17)  21121 型 .....   20
( 18)  51121 型 .....   60
( 19)  12121 型 .....   20
( 20)  22121 型 .....   20
( 21)  52121 型 .....   60
( 22)  14121 型 .....   60
( 23)  24121 型 .....   60
( 24)  44121 型 .....   60
( 25)  54121 型 .....  120
( 26)  11221 型 .....   20
( 27)  21221 型 .....   20
( 28)  51221 型 .....   60
( 29)  12221 型 .....   20
( 30)  22221 型 .....   20
( 31)  52221 型 .....   60
( 32)  14221 型 .....   60
( 33)  24221 型 .....   60
( 34)  44221 型 .....   60
( 35)  54221 型 .....  120
( 36)  11321 型 .....   60
( 37)  21321 型 .....   60
( 38)  31321 型 .....   60
( 39)  51321 型 .....  120
( 40)  12321 型 .....   60
( 41)  22321 型 .....   60
( 42)  32321 型 .....   60
( 43)  52321 型 .....  120
( 44)  13321 型 .....   60
( 45)  23321 型 .....   60
( 46)  33321 型 .....   60
( 47)  53321 型 .....  120
( 48)  14321 型 .....  120
( 49)  24321 型 .....  120
( 50)  34321 型 .....  120
( 51)  44321 型 .....  120
( 52)  54321 型 .....  120
..............................
  計                     3125
  源氏香図総数 =   52
```

4．源氏香図の一般化

源氏香の香数を5種と限定せずに，一般にn種の香を用いて，各香をn包ずつ用意した合計 n^2 包の中から任意にn包取り出して並べ，それらの異種同種を判定するゲームを考え，これをn香の源氏香と呼ぶことにする．とり出したn包にn本の縦線を順に対応させ，同種の香包に対応する縦線の上端を横線で結ぶことによってできる図形の総数を$f(n)$で示そう．このとき，「n種以上の香を用い，各香をn包以上用意した中からn包とり出して並べる」としても，これらを表す図形の総数は，とり出す香包数nのみによって決まり，やはり$f(n)$である．それは図形の作り方を考えれば明らかである．

さて，$n \leqq 5$ のとき，直接$f(n)$を求めれば

n	1	2	3	4	5
図形					源氏香図
$f(n)$	1	2	5	15	52

となる（実際に図形を作ってみた結果）．

いま，$f(0)=1$ と定義すれば，次の等式は簡単に検証できるであろう．

$f(2)={}_1C_0f(0)+{}_1C_1f(1)$

$f(3)={}_2C_0f(0)+{}_2C_1f(1)+{}_2C_2f(2)$

$f(4)={}_3C_0f(0)+{}_3C_1f(1)+{}_3C_2f(2)+{}_3C_3f(3)$

$f(5)={}_4C_0f(0)+{}_4C_1f(1)+{}_4C_2f(2)+{}_4C_3f(3)$
$\qquad +{}_4C_4f(4)$

一般に，2以上の整数nに対して次の漸化式を証明することができる．

（定理） $f(n)=\sum_{r=0}^{n-1}{}_{n-1}C_rf(r)$ $(n \geqq 2)$

ただし，$f(0)=1$，$f(1)=1$ である．

（証明） $n=2$ のときには，上の等式が成立するので，2以上，n以下のすべての正整数について上の等式が成り立つと仮定しよう．

$n+1$種の香をa_1，a_2，……，a_n，a_{n+1}とする（各香は，それぞれ$n+1$包）．この中から同香を許して$n+1$包の香を選んで，$n+1$個の場所に右から順に1列に並べる場合に，最後の $n+1$番目の香と同種のものが，その前にr個 $(0 \leqq r \leqq n)$ あったとする．r個の選び方は${}_nC_r$通りあり，それらの各々の場合に，$n+1$番目とは異なる **$(n-r)$ 包の香による香図**数は，仮定から求められて$f(n-r)$である．

よって，この場合の源氏香図は${}_nC_rf(n-r)$ 通りできる．$0 \leqq r \leqq n$，${}_nC_r={}_nC_{n-r}$ であるから，



$$f(n+1)=\sum_{r=0}^{n}{}_nC_rf(n-r)=\sum_{r=0}^{n}{}_nC_rf(r) \qquad \text{（証終）}$$

この定理の漸化式によって，パソコンを用いて$f(n)$の値を求めることは容易であり，次の結果が得られました．ただし，$n \leqq 15$

```
10   ' n 香の源氏香図の総数計算
20   DIM F#(15),C#(15,15)
30   FOR M=1 TO 15
40   C#(M,0)=1:C#(M,M)=1
50   NEXT M
60   FOR M=2 TO 15
70     FOR N=1 TO M-1
80     C#(M,N)=C#(M-1,N-1)+C#(M-1,N)
90     NEXT N
100  NEXT M
110  F#(0)=1:F#(1)=1
120    FOR K=2 TO 15
130     FOR I= 0 TO K
140      S#=S#+C#(K-1,I)*F#(I)
150     NEXT I
160      F#(K)=S#:S#=0
180    NEXT K
190   LPRINT "   n          f(n)    "
195   FOR K=1 TO 15
200   LPRINT USING "####    ##########  ";K;F#(K)
210   NEXT K
220   END
```

n	f(n)
1	1
2	2
3	5
4	15
5	52
6	203
7	877
8	4140
9	21147
10	115975
11	678570
12	4213597
13	27644437
14	190899322
15	1382958545

この表から，ゲームとしての源氏香図は，$n=5$ の場合が適切であったことがわかる．

5．終わりに

源氏香図の一般化は私の創意ではなく，約 30 年程前に都立小金井工高教諭田沼一郎氏（故人）は，順列組合せの考え方で直接に n 香の源氏香図の総数 $f(n)$ を計算し，$f(n)$ の式から逆に漸化式を少し違った形に導かれています．

田沼一郎氏の 2 定理：

（定理 1）

$$
\begin{aligned}
f(n)=2^n-n&+\frac{1}{2!}\binom{n}{2}\binom{n-2}{2}\left\{1+\frac{1}{3}\binom{n-4}{2}\right.\\
&\left.+\frac{1}{3\cdot4}\binom{n-4}{2}\binom{n-6}{2}+\cdots\right\}\\
&+\binom{n}{3}\binom{n-3}{2}\left\{1+\frac{1}{2!}\binom{n-5}{2}\right.\\
&\left.+\frac{1}{3!}\binom{n-5}{2}\binom{n-7}{2}+\cdots\right\}\\
&+\frac{1}{2!}\binom{n}{3}\binom{n-3}{3}\left\{1+\binom{n-6}{2}\right.\\
&\left.+\frac{1}{2!}\binom{n-6}{2}\binom{n-8}{2}+\cdots\right\}+\cdots\\
&+\binom{n}{n-3}\binom{3}{2}+\binom{n}{n-3}\binom{3}{3}\\
&+\binom{n}{n-2}\binom{2}{2}
\end{aligned}
$$

ただし，$\binom{p}{q}={}_pC_q$，$(q\leqq p)$；$\binom{p}{q}=0$，$(q>p)$

（定理 2） $f(1)=1$，$f(2)=2$ で

$$f(n)=\sum_{r=0}^{n-2}\binom{n-2}{r}f(r+1)+f(n-1),\quad (n\geqq 3)$$

（東京都立 神代高等学校）

確率・統計　　**場合の数**

［問題］　a，b，c，d，e の 5 文字から，同じ文字を繰り返し用いることを許して 5 文字を並べる．
このとき，a b c d e　なら 　　a a b c b　なら 
d e e d d　なら 　　b b b b b　なら 　……
というように，5 文字に 5 本の縦線を順番に対応させて，同じ文字があると，対応する縦線を横線で結んだ図形を作っていくことにすれば，異なる図形は何通りできるか．

［ヒント］　次のように場合分けをして考えるとよい．

i）同じ文字がない場合　　ii）2 個だけが同じ文字の場合
iii）2 個ずつ同じ文字が 2 組ある場合　　iv）3 個だけが同じ文字の場合
v）3 個が同じ文字，残りも別の同じ文字である場合
vi）4 個が同じ文字の場合　　vii）5 個とも同じ文字の場合

源氏香図

「（参考資料）日本国語大辞典 小学館」